*-------------------------------------------------------------------------------*;

*----------------- %unisum: --- Univariate Data Summary Macro ------------------*;

*-------------------------------------------------------------------------------*;

/*

Macro %UNISUM read_me notes:

Inputs:

indata


Name of the SAS dataset input to analyze

outdata


Name of the SAS dataset to output results to

var_list

A list of variable names in "indata" to analyze. 





Must all be numeric variable.

by



Optional list of variable names in "indata" to analyze by.





Can be numeric or character, formatted or unformatted.





Not required to input.

output


A list of statistics and names to output. 





Must follow formatting for proc univariate's output statement.





Such as: mean=output_mean n=output_n median=output_median

print


Option to suppress automatic printed output. 





Set to "Y" for "yes" as default, to print the output.





Set to anything besides Y to surpress printing.





Not required to input.

Returns:

Creates and prings (to SAS Listing) a dataset with the names of the analysis variables,

requested output statistics and any option by variables.

Example calls:

--This example has no by variable and defaults to printing:

%unisum(indata=dummy_data, outdata=out1, var_list=var1 var2, output= mean=mean n=n median=median);

run;

--This example includes by variables and suppresses the automatic printing:

%unisum(indata=dummy_data, outdata=out1, var_list=var1 var2, by=factor1 factor2, output= mean=mean n=n median=median, print=n);

run;

Other notes:

--Uses the macro %words which is also included in this file.

*/

*-------------------------------------------------------------------------------*;

*---------------------------- WORDS macro -- START -----------------------------*;

/**

Macro %WORDS read_me notes:

Inputs:

string

A character string made up of multiple "words" each separated by blank space. 




No commas!

Returns:

Returns only a count of the number of "words" in the character string

Example calls:

Used within another macro to assign a local macro variable the number of "words" in a character string.

%let num_var=%words(&var_list);

This creates the macro variable "num_var" which is the number of words(variables) in the macro input 

variable &var_list.

**/

*** MACRO COUNTS NUMBER OF WORDS IN A STRING ***;

%macro words(string);

    %local count word;

    %let count=1;

    %let word=%qscan(&string,&count);

    %do %while(&word ne);

      %let count=%eval(&count+1);

      %let word=%qscan(&string,&count);

    %end;

    %eval(&count-1)

%mend words;

run;

*---------------------------- WORDS macro ---- END -----------------------------*;

*-------------------------------------------------------------------------------*;

*-------------------------------------------------------------------------------*;

*--------------------------- UNISUM macro --- START ----------------------------*;

%macro unisum(indata=, outdata=, var_list=, by=, output=, print=Y);

*------Create macro variables for all the variables to analyze------*;

*Get number of words in var_list;

%let num_var=%words(&var_list);

*Create a macro variable for each class variable;

%do i=1 %to &num_var;

  %let var&i=%scan(&var_list,&i);

%end;

*------Create macro variables for all the variables to analyze------*;

*------Optional sort------*;

%if &by^= %then %do;

proc sort data=&indata out=_tempdata1; by &by;

run;

%end;

%else %do;

data _tempdata1;

set &indata;

run;

%end;

*------Optional sort------*;

*------Null dataset to combine with output datasets------*;

data &outdata;

set _null_;

run;

*------Null dataset to combine with output datasets------*;

*------Macro loop running proc univariate for each variable, produce output------*;

run; ods listing close; run;

%do i=1 %to &num_var;

*--Proc univariate to get output--*;

proc univariate data=_tempdata1;

by &by;

var &&var&i;

output out=_tempout&i &output;

run;

*--Add variable name to dataset--*;

data _tempout&i;

set _tempout&i;

length variable $ 20;

variable="&&var&i";

run;

*--Concatenate current dataset onto final output dataset--*;

data &outdata;

set &outdata _tempout&i;

run;

%end;

run; ods listing; run;

*------Macro loop running proc univariate for each variable, produce output------*;

*------Optional printed output------*;

%if &print=Y %then %do;

proc print data=&outdata;

run;

%end;

*------Optional printed output------*;

*------Delete temporary working datasets in macro------*;

proc datasets library=work nolist nowarn;

delete _tempdata1 _tempout1-_tempout&num_var;

run;

quit;

*------Delete temporary working datasets in macro------*;

run;

%mend;

*--------------------------- UNISUM macro ----- END ----------------------------*;

*-------------------------------------------------------------------------------*;
